Pensieve Header: The Polyak Algebra.

2.5. **The Polyak Algebra.** The *Polyak algebra* is the quotient of $\mathcal{A}$ by the following relations:

(7)



(8)



(9)



R3b

Program and Testing

Drawing Arrow Diagrams

```
SetAttributes[Diag, Orderless];
Draw[expr_] := expr /. diag_Diag ⧴ Draw[diag];
Draw[diag_Diag] := Module[
   {n = Length[diag]},
   Graphics[{
     Line[{{0, 0}, {2 n + 1, 0}}],
     (List @@ diag) /.  {
       Ar[i_, j_] ⧴ {
         BezierCurve[{
           {i, 0}, {(i + j) / 2, Abs[i - j]}, {j, 0}
          }],
         Line[{{j - 0.2 + 0.2 Sign[i - j], 0.4}, {j, 0}, {j + 0.2 + 0.2 Sign[i - j], 0.4}}]
        }
      }
    }
   ]
  ];
```

```
Draw[Diag[Ar[3, 1], Ar[2, 4]]]
```



Generating Arrow Diagrams

```
Place[{Ar}, {i_, j_}] := {Diag[Ar[i, j]], Diag[Ar[j, i]]};
Place[{Ar, objs__}, {i_, rest__}] := Flatten[Table[
    Outer[Join,
     Place[{Ar}, {i, {rest}[[k]]}],
     Place[{objs}, Delete[{rest}, k]]
    ],
    {k, Length[{rest}]}
   ]];
Diagrams[k_. * Ar] := Place[Table[Ar, {k}], Range[2 k]];
```

```
Diagrams[2 Ar]
```

```
{Diag[Ar[1, 2], Ar[3, 4]], Diag[Ar[1, 2], Ar[4, 3]], Diag[Ar[2, 1], Ar[3, 4]],
 Diag[Ar[2, 1], Ar[4, 3]], Diag[Ar[1, 3], Ar[2, 4]], Diag[Ar[1, 3], Ar[4, 2]],
 Diag[Ar[2, 4], Ar[3, 1]], Diag[Ar[3, 1], Ar[4, 2]], Diag[Ar[1, 4], Ar[2, 3]],
 Diag[Ar[1, 4], Ar[3, 2]], Diag[Ar[2, 3], Ar[4, 1]], Diag[Ar[3, 2], Ar[4, 1]]}
```

```
Draw[Diagrams[2 Ar]]
```

{  ,

 ,

 ,

 ,

 ,





,



,



,







The "Round" Case

```
DiagramRotateLeft[d_Diag] := Module[
   {labels = Union @@ (List @@ Apply[List, d, {1}])},
   d /. Thread[Rule[labels, RotateLeft[labels]]]
  ];
DiagramRotateToMinimal[expr_] := expr /. d_Diag :> Module[
     {bd = d, rd = DiagramRotateLeft[d]},
     While[rd =!= d,
      bd = First[Sort[{bd, rd}]];
      rd = DiagramRotateLeft[rd]];
     bd
    ];
```

```
Union[DiagramRotateToMinimal[Diagrams[2 Ar]]]
```

```
{Diag[Ar[1, 2], Ar[3, 4]], Diag[Ar[1, 2], Ar[4, 3]],
 Diag[Ar[1, 3], Ar[2, 4]], Diag[Ar[1, 4], Ar[3, 2]]}
```

```
Draw[Union[DiagramRotateToMinimal[Diagrams[2 Ar]]]]
```



Generating Relators

```
Place[{r : (TC | R3b), objs__}, {i_, rest__}] := Flatten[Table[
    Outer[Join,
     Place[{r}, {i, {rest}[[j]], {rest}[[k]]}],
     Place[{objs}, Delete[{rest}, {{j}, {k}}]]
    ],
    {k, 2, Length[{rest}]}, {j, 1, k - 1}
   ]];
Place[{R3b}, {i_, j_, k_}] :=
  Permutations[{i, j, k}] /. {i1_, j1_, k1_} :> Diag[R3b[i1, j1, k1]];
Diagrams[R3b] := Place[{R3b}, {1, 2, 3}];
Diagrams[R3b + k_. * Ar] /; k > 0 := Flatten[
   Place[#, Range[2 k + 3]] & /@ Permutations[Table[Ar, {k}] ~Append~ R3b]
  ];
Place[{TC}, {i_, j_, k_}] := Diag /@ {TC[i, j, k], TC[j, k, i], TC[k, i, j]};
Diagrams[TC] := Place[{TC}, {1, 2, 3}];
Diagrams[TC + k_. * Ar] /; k > 0 := Flatten[
   Place[#, Range[2 k + 3]] & /@ Permutations[Table[Ar, {k}] ~Append~ TC]
  ];
Diagrams[TC + k_. * Ar] /; k < 0 := {};
Place[{R2c}, {i_, j_}] := {Diag[R2c[i, j]], Diag[R2c[j, i]]};
Place[{r : (R2c), objs__}, {i_, rest__}] := Flatten[Table[
    Outer[Join,
     Place[{r}, {i, {rest}[[j]]}],
     Place[{objs}, Delete[{rest}, j]]
    ],
    {j, 1, Length[{rest}]}
   ]];
Diagrams[R2c] := Place[{R2c}, {1, 2}];
Diagrams[R2c + k_. * Ar] /; k > 0 := Flatten[
   Place[#, Range[2 k + 2]] & /@ Permutations[Table[Ar, {k}] ~Append~ R2c]
  ];
Diagrams[R2c + k_. * Ar] /; k < 0 := {};
Place[{R1d}, {i_}] := {Diag[R1d[i]]};
Place[{R1a}, {i_}] := {Diag[R1a[i]]};
Place[{r : (R1d | R1a), objs__}, {i_, rest__}] := Flatten[
   Outer[Join,
    Place[{r}, {i}],
    Place[{objs}, {rest}]
   ]];
Diagrams[R1d] := Place[{R1d}, {1}];
Diagrams[R1a] := Place[{R1a}, {1}];
Diagrams[R1d + k_. * Ar] /; k > 0 := Flatten[
   Place[#, Range[2 k + 1]] & /@ Permutations[Table[Ar, {k}] ~Append~ R1d]
  ];
Diagrams[R1a + k_. * Ar] /; k > 0 := Flatten[
   Place[#, Range[2 k + 1]] & /@ Permutations[Table[Ar, {k}] ~Append~ R1a]
  ];
Diagrams[R1d + k_. * Ar] /; k < 0 := {};
Diagrams[R1a + k_. * Ar] /; k < 0 := {};
```

```
Take[Diagrams[R3b + 1 Ar], 5]
```

```
{Diag[Ar[1, 2], R3b[3, 4, 5]], Diag[Ar[1, 2], R3b[3, 5, 4]],
 Diag[Ar[1, 2], R3b[4, 3, 5]], Diag[Ar[1, 2], R3b[4, 5, 3]], Diag[Ar[1, 2], R3b[5, 3, 4]]}
```

```
{Diagrams[R1d + Ar], Diagrams[R1a + Ar]}
```

```
{{Diag[Ar[1, 2], R1d[3]], Diag[Ar[2, 1], R1d[3]], Diag[Ar[1, 3], R1d[2]],
  Diag[Ar[3, 1], R1d[2]], Diag[Ar[2, 3], R1d[1]], Diag[Ar[3, 2], R1d[1]]},
 {Diag[Ar[1, 2], R1a[3]], Diag[Ar[2, 1], R1a[3]], Diag[Ar[1, 3], R1a[2]],
  Diag[Ar[3, 1], R1a[2]], Diag[Ar[2, 3], R1a[1]], Diag[Ar[3, 2], R1a[1]]}}
```

Generating Relations

```
NormalizeDiag[diag_Diag] := Module[
   {indices = Union @@ (List @@ diag /. Ar → List)},
   diag /. Thread[indices → Range[Length[indices]]]
  ];
R[Diag[lft___, R3b[i_, j_, k_], rgt___]] := (
   (* RHS of (9) *)
   NormalizeDiag[Diag[lft, Ar[i, j], Ar[i + 0.5, k], Ar[j + 0.5, k + 0.5], rgt]]
    + NormalizeDiag[Diag[lft, Ar[i, j], Ar[i + 0.5, k], rgt]]
    + NormalizeDiag[Diag[lft, Ar[i, j], Ar[j + 0.5, k], rgt]]
    + NormalizeDiag[Diag[lft, Ar[i, k], Ar[j, k + 0.5], rgt]]
    (* LHS of (9) *)
    - NormalizeDiag[Diag[lft, Ar[j, k], Ar[i, k + 0.5], Ar[i + 0.5, j + 0.5], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, k], Ar[i + 0.5, j], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, j + 0.5], Ar[j, k], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, k + 0.5], Ar[j, k], rgt]]
  );
RWrong1[Diag[lft___, R3b[i_, j_, k_], rgt___]] := (
   (* RHS of (9) *)
   NormalizeDiag[Diag[lft, Ar[i, j], Ar[i + 0.5, k], Ar[j + 0.5, k + 0.5], rgt]]
    + NormalizeDiag[Diag[lft, Ar[i, j], Ar[i + 0.5, k], rgt]]
    + NormalizeDiag[Diag[lft, Ar[i, j], Ar[j + 0.5, k], rgt]]
    + NormalizeDiag[Diag[lft, Ar[i, k], Ar[j, k + 0.5], rgt]]
    (* LHS of (9) *)
    + NormalizeDiag[Diag[lft, Ar[j, k], Ar[i, k + 0.5], Ar[i + 0.5, j + 0.5], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, k], Ar[i + 0.5, j], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, j + 0.5], Ar[j, k], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, k + 0.5], Ar[j, k], rgt]]
  );
RWrong2[Diag[lft___, R3b[i_, j_, k_], rgt___]] := (
   (* RHS of (9) *)
   NormalizeDiag[Diag[lft, Ar[i, j], Ar[i + 0.5, k], Ar[j + 0.5, k + 0.5], rgt]]
    + NormalizeDiag[Diag[lft, Ar[i, j], Ar[i + 0.5, k], rgt]]
    + NormalizeDiag[Diag[lft, Ar[i, j], Ar[j + 0.5, k], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, k], Ar[j, k + 0.5], rgt]]
    (* LHS of (9) *)
    - NormalizeDiag[Diag[lft, Ar[j, k], Ar[i, k + 0.5], Ar[i + 0.5, j + 0.5], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, k], Ar[i + 0.5, j], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, j + 0.5], Ar[j, k], rgt]]
    - NormalizeDiag[Diag[lft, Ar[i, k + 0.5], Ar[j, k], rgt]]
  );
```

```
RMinus[n_, Diag[lft___, R3b[i_, j_, k_], rgt___]] := Module[
   {Expand2, Expand3, n1},
   n1 = n - Length[{lft, rgt}];
   Expand2[Diag[Ar[i1_, j1_], Ar[i2_, j2_]]] := Sum[
     (-1)^(k1 + k2) * NormalizeDiag[Flatten[Diag[
         lft,
         Diag @@ Table[Ar[i1 + 0.01 q, j1 + 0.01 q], {q, k1}],
         Diag @@ Table[Ar[i2 + 0.01 q, j2 + 0.01 q], {q, k2}],
         rgt
        ]]],
     {k1, n1 - 1}, {k2, n1 - k1}
    ];
   Expand3[Diag[Ar[i1_, j1_], Ar[i2_, j2_], Ar[i3_, j3_]]] := Sum[
     (-1)^(k1 + k2 + k3) * NormalizeDiag[Flatten[Diag[
         lft,
         Diag @@ Table[Ar[i1 + 0.01 q, j1 + 0.01 q], {q, k1}],
         Diag @@ Table[Ar[i2 + 0.01 q, j2 + 0.01 q], {q, k2}],
         Diag @@ Table[Ar[i3 + 0.01 q, j3 + 0.01 q], {q, k3}],
         rgt
        ]]],
     {k1, n1 - 2}, {k2, n1 - k1 - 1}, {k3, n1 - k1 - k2}
    ];
   Expand[Plus[
     (
       Diag[Ar[i, j], Ar[i + 0.5, k]] +
        Diag[Ar[i, j], Ar[j + 0.5, k]] + Diag[Ar[i, k], Ar[j, k + 0.5]]
        - Diag[Ar[i, k], Ar[i + 0.5, j]] - Diag[Ar[i, j + 0.5], Ar[j, k]] -
        Diag[Ar[i, k + 0.5], Ar[j, k]]
      ) /. d_Diag :> Expand2[d],
     (Diag[Ar[i, j], Ar[i + 0.5, k], Ar[j + 0.5, k + 0.5]] -
        Diag[Ar[j, k], Ar[i, k + 0.5], Ar[i + 0.5, j + 0.5]]) /. d_Diag :> Expand3[d]
    ]]
  ];
R[n_, Diag[lft___, R2c[i_, j_], rgt___]] := Module[
   {n1},
   n1 = n - Length[{lft, rgt}];
   Sum[
    Expand[(-1)^k * s * NormalizeDiag[Flatten[Diag[
         lft,
         Diag @@ Table[Ar[i + 0.01 q, j + s * 0.01 q], {q, k}],
         rgt
        ]]]],
```

```
    {k, 2, n1}, {s, -1, 1, 2}
   ]
  ];
R[n_, Diag[lft___, R1d[i_], rgt___]] := Module[
   {n1},
   n1 = n - Length[{lft, rgt}];
   Sum[
    Expand[(-1)^k * NormalizeDiag[Flatten[Diag[
          lft,
          Diag @@ Table[Ar[i + 0.01 q, i + 0.5 + 0.01 q], {q, k}],
          rgt
         ]]]],
    {k, 2, n1}
   ]
  ];
R[n_, Diag[lft___, R1a[i_], rgt___]] := Module[
   {n1},
   n1 = n - Length[{lft, rgt}];
   Sum[
    Expand[(-1)^k * NormalizeDiag[Flatten[Diag[
          lft,
          Diag @@ Table[Ar[i + 0.5 + 0.01 q, i + 0.01 q], {q, k}],
          rgt
         ]]]],
    {k, 2, n1}
   ]
  ];
R[Diag[lft___, TC[i_, j_, k_], rgt___]] := (
   + NormalizeDiag[Diag[lft, Ar[k, i], Ar[k + 0.5, j], rgt]]
    - NormalizeDiag[Diag[lft, Ar[k + 0.5, i], Ar[k, j], rgt]]
  );
ContainsShortArrow[diag_Diag] := (1 == Min[List @@ diag /. Ar[i_, j_] :> Abs[i - j]]);
DescendingQ[diag_Diag] := And @@ (OrderedQ /@ diag);

Draw[R /@ Take[Diagrams[R3b + 1 Ar], 2]]
```

{  +

 −

2  +

 −


,


−


+


−



Producing "Quotient Spaces"

```
Options[PSpace] = {Debug → False, Gr → False, ReduceEquivalences → True};
PSpace[m_] /; m < 2 := QuotientSpace[Diagrams[m Ar], {}];
PSpace[m_, reltypes_List, opts___] /; m ≥ 2 := Module[
  {
   debug = Debug /. {opts} /. Options[PSpace],
   gr = Gr /. {opts} /. Options[PSpace],
   reduceequivalences = ReduceEquivalences /. {opts} /. Options[PSpace],
   diags, rels, i, filter, equalities, equivclasses, pos, redrule
```

```
},
If[debug, Print[Date[], ": Starting work..."]];
diags = Join @@ Table[Diagrams[k Ar], {k, If[gr, m, 1], m}];
filter = Identity;
reltypes /. {
  "R1" :> (
    diags = Select[diags, ! ContainsShortArrow[#] &]
   ),
  "Descending" :> (
    filter = Select[#, DescendingQ] &;
    diags = Select[diags, DescendingQ]
   )
 };
If[debug, Print[Date[], ": Computed ", Length[diags], " diagrams..."]];
rels = Join @@ (reltypes /. {
     "R1" → {},
     "Descending" → {},
     "Round" :> (diags /.
        d_Diag :> (NormalizeDiag[d /. {1 → 2 Length[d], i_Integer :> i - 1}] - d)
      ),
     "R3b" :> (
       (R /@ Join @@ Table[Diagrams[R3b + k Ar] // filter, {k, If[gr, m - 2, 0], m - 2}]) /.
        diag_Diag /; Length[diag] > m :> 0
      ),
     "RWrong1" :> (
       (RWrong1 /@ Join @@ Table[Diagrams[R3b + k Ar], {k, If[gr, m - 2, 0], m - 2}]) /.
        diag_Diag /; Length[diag] > m :> 0
      ),
     "RWrong2" :> (
       (RWrong2 /@ Join @@ Table[Diagrams[R3b + k Ar], {k, If[gr, m - 2, 0], m - 2}]) /.
        diag_Diag /; Length[diag] > m :> 0
      ),
     "R3bMinus" :> (
       RMinus[m, #] & /@
        Join @@ Table[Diagrams[R3b + k Ar] // filter, {k, If[gr, m - 2, 0], m - 2}]
      ),
     "R2c" :> (
       R[m, #] & /@
        Join @@ Table[Diagrams[R2c + k Ar] // filter, {k, If[gr, m - 2, 0], m - 2}]
      ),
     "R1d" :> (
```

```
          R[m, #] & /@
           Join @@ Table[Diagrams[R1d + k Ar] // filter, {k, If[gr, m - 2, 0], m - 2}]
         ),
        "R1a" :> (
          R[m, #] & /@
           Join @@ Table[Diagrams[R1a + k Ar] // filter, {k, If[gr, m - 2, 0], m - 2}]
         ),
        "TC" :> (
          R /@ Join @@ Table[Diagrams[TC + k Ar], {k, If[gr, m - 2, 0], m - 2}]
         )
       });
  reltypes /. {
    "R1" :> (
      rels = DeleteCases[rels /. {d_Diag /; ContainsShortArrow[d] :> 0}, 0]
     )
   };
  If[debug, Print[Date[], ": Computed ", Length[rels], " relations..."]];
  If[reduceequivalences,
   equalities = Cases[rels, d1_Diag - d2_Diag :> {d1, d2}];
   equivclasses = Fold[
     (
       pos = First /@ Position[#1, #2];
       Append[Delete[#1, List /@ pos], Union @@ (#1[[pos]])]
      ) &,
     equalities,
     Union @@ equalities
    ];
   redrule = Dispatch[Flatten[Table[
       class = equivclasses[[i]];
       (# → First[class]) & /@ Rest[class],
       {i, Length[equivclasses]}
      ]]];
   diags = Union[diags /. redrule];
   rels = Union[rels /. redrule];
  ];
  QuotientSpace[diags, rels]
 ]

Draw[PSpace[2, {"R3b", "R2c"}]]
```

QuotientSpace[{


,


,


,


,


,


,



,



,



,



,

,



,



,





2 ,



–

+















+



−



′









2 ,











Computing Dimensions

```
Options[Dim] = {Debug → False, UseLinBox → False};
Dim[QuotientSpace[diags_List, rels_List], opts___] := Module[
   {
    debug = Debug /. {opts} /. Options[Dim],
    uselinbox = UseLinBox /. {opts} /. Options[Dim],
    diagtoindex, mat, rel, matfile, rank
   },
   diagtoindex = Dispatch[Thread[Rule[diags, Range[Length[diags]]]]];
   If[uselinbox,
    matfile = OpenWrite["matfile.matrix"];
    WriteString[matfile, Length[rels], " ", Length[diags], " S\n"];
    Do[
     rel = rels[[i]];
     WriteString[matfile, StringJoin @@ Flatten[{
          ToString[Count[{rel}, _Diag, Infinity]], " ",
          {
             ToString[(# /. diagtoindex) - 1], " ",
             ToString[Coefficient[rel, #]], " "
           } & /@ Cases[{rel}, diag_Diag, Infinity],
          "\n"
         }]],
     {i, Length[rels]}
    ];
    matfile = Close[matfile];
    rank = RunThrough["./ComputeRank " <> matfile, Random[][[2]];
    Length[diags] - rank,
    (* else *) mat = SparseArray[
      Join @@ Table[
        rel = rels[[i]];
        {i, # /. diagtoindex} → Coefficient[rel, #] & /@
         Cases[{rel}, diag_Diag, Infinity],
        {i, Length[rels]}
       ],
      {Length[rels], Length[diags]}
     ];
    If[debug, Print[Date[], ": Computed mat..."]];
    Length[diags] - MatrixRank[mat]
   ]
  ];
DimP[m_, reltypes_List, opts___] := Dim[PSpace[m, reltypes, opts], opts]
```

Analysis

```
Analyze[m_Integer, reltypes_List, opts___Rule] := Module[
   {qs},
   Table[
    {
     qs = PSpace[n, reltypes, Gr → True, opts];
     (Length /@ qs) → Dim[qs, opts],
     qs = PSpace[n, reltypes, Gr → False, opts];
     (Length /@ qs) → Dim[qs, opts]
    },
    {n, 2, m}
   ]
  ];
Analyze[reltypes_List, opts___Rule] := Analyze[4, reltypes, opts];
```

Runs

Runs as in the paper

```
Analyze[{"R3b", "R2c", "R1", "Round"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[1, 3] \to 0 & \text{QuotientSpace}[1, 3] \to 0 \\ \text{QuotientSpace}[12, 29] \to 1 & \text{QuotientSpace}[12, 31] \to 1 \\ \text{QuotientSpace}[107, 297] \to 4 & \text{QuotientSpace}[118, 336] \to 5 \end{pmatrix}$$

```
Analyze[{"R3b", "R2c", "R1"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[4, 4] \to 2 & \text{QuotientSpace}[4, 4] \to 2 \\ \text{QuotientSpace}[32, 37] \to 7 & \text{QuotientSpace}[34, 39] \to 9 \\ \text{QuotientSpace}[460, 749] \to 42 & \text{QuotientSpace}[500, 829] \to 51 \end{pmatrix}$$

```
Analyze[{"R3b", "R2c", "R1", "Descending"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[1, 1] \to 0 & \text{QuotientSpace}[1, 1] \to 0 \\ \text{QuotientSpace}[3, 3] \to 1 & \text{QuotientSpace}[3, 4] \to 1 \\ \text{QuotientSpace}[22, 23] \to 6 & \text{QuotientSpace}[27, 33] \to 7 \end{pmatrix}$$

```
Analyze[{"R3b", "R2c", "Round"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[3, 3] \to 1 & \text{QuotientSpace}[4, 3] \to 2 \\ \text{QuotientSpace}[16, 21] \to 2 & \text{QuotientSpace}[21, 24] \to 4 \\ \text{QuotientSpace}[160, 329] \to 7 & \text{QuotientSpace}[187, 362] \to 11 \end{pmatrix}$$

```
Analyze[{"R3b", "R2c"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[10, 7] \to 5 & \text{QuotientSpace}[12, 7] \to 7 \\ \text{QuotientSpace}[96, 121] \to 15 & \text{QuotientSpace}[110, 129] \to 22 \\ \text{QuotientSpace}[1332, 2401] \to 67 & \text{QuotientSpace}[1466, 2553] \to 89 \end{pmatrix}$$

```
Analyze[{"R3b", "R2c", "Descending"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[2, 2] \to 1 & \text{QuotientSpace}[3, 2] \to 2 \\ \text{QuotientSpace}[9, 11] \to 2 & \text{QuotientSpace}[13, 13] \to 4 \\ \text{QuotientSpace}[63, 96] \to 8 & \text{QuotientSpace}[82, 114] \to 12 \end{pmatrix}$$

```
Analyze[{"R3b", "R1", "R1d", "R1a", "Round"}] // MatrixForm
```

```
{{QuotientSpace[1, 3] → 0, QuotientSpace[1, 3] → 0},
 {QuotientSpace[12, 29] → 1, QuotientSpace[12, 31] → 1},
 {QuotientSpace[107, 297] → 4, QuotientSpace[118, 334] → 5}}
```

```
Analyze[{"R3b", "R1d", "R1a", "R1"}] // MatrixForm
```

```
{{QuotientSpace[4, 4] → 2, QuotientSpace[4, 4] → 2},
 {QuotientSpace[32, 37] → 7, QuotientSpace[34, 39] → 9},
 {QuotientSpace[460, 749] → 42, QuotientSpace[500, 821] → 51}}
```

```
Analyze[{"R3b", "R1d", "R1", "Descending"}] // MatrixForm
```

```
{{QuotientSpace[1, 1] → 0, QuotientSpace[1, 1] → 0},
 {QuotientSpace[3, 3] → 1, QuotientSpace[3, 4] → 1},
 {QuotientSpace[22, 23] → 6, QuotientSpace[27, 31] → 7}}
```

```
Analyze[{"R3b", "Round"}] // MatrixForm
```

```
{{QuotientSpace[4, 3] → 2, QuotientSpace[5, 3] → 3},
 {QuotientSpace[22, 21] → 5, QuotientSpace[27, 23] → 8},
 {QuotientSpace[218, 353] → 19, QuotientSpace[245, 379] → 27}}
```

```
Analyze[{"R3b"}] // MatrixForm
```

```
{{QuotientSpace[12, 6] → 7, QuotientSpace[14, 6] → 9},
 {QuotientSpace[120, 120] → 27, QuotientSpace[134, 126] → 36},
 {QuotientSpace[1680, 2520] → 139, QuotientSpace[1814, 2646] → 175}}
```

```
Analyze[{"R3b", "Descending"}] // MatrixForm
```

```
{{QuotientSpace[3, 1] → 2, QuotientSpace[4, 1] → 3},
 {QuotientSpace[15, 10] → 6, QuotientSpace[19, 11] → 9},
 {QuotientSpace[105, 105] → 24, QuotientSpace[124, 116] → 33}}
```

```
Analyze[{"R2c", "R1", "Round"}] // MatrixForm
```

```
{{QuotientSpace[1, 2] → 0, QuotientSpace[1, 2] → 0},
 {QuotientSpace[12, 13] → 4, QuotientSpace[12, 13] → 4},
 {QuotientSpace[107, 75] → 44, QuotientSpace[118, 82] → 48}}
```

```
Analyze[{"R2c", "R1"}] // MatrixForm
```

```
{{QuotientSpace[4, 2] → 2, QuotientSpace[4, 2] → 2},
 {QuotientSpace[32, 5] → 28, QuotientSpace[34, 5] → 30},
 {QuotientSpace[460, 41] → 420, QuotientSpace[500, 51] → 450}}
```

```
Analyze[{"R2c", "R1", "Descending"}] // MatrixForm
```

```
{{QuotientSpace[1, 1] → 0, QuotientSpace[1, 1] → 0},
 {QuotientSpace[3, 2] → 2, QuotientSpace[3, 2] → 2},
 {QuotientSpace[22, 5] → 18, QuotientSpace[27, 8] → 20}}
```

```
Analyze[{"R2c", "Round"}] // MatrixForm
```

```
{{QuotientSpace[3, 1] → 3, QuotientSpace[4, 1] → 4},
 {QuotientSpace[16, 1] → 16, QuotientSpace[21, 2] → 20},
 {QuotientSpace[160, 1] → 160, QuotientSpace[187, 8] → 180}}
```

```
Analyze[{"R2c"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[10, 1] \to 10 & \text{QuotientSpace}[12, 1] \to 12 \\ \text{QuotientSpace}[96, 1] \to 96 & \text{QuotientSpace}[110, 3] \to 108 \\ \text{QuotientSpace}[1332, 1] \to 1332 & \text{QuotientSpace}[1466, 27] \to 1440 \end{pmatrix}$$

```
Analyze[{"R2c", "Descending"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[2, 1] \to 2 & \text{QuotientSpace}[3, 1] \to 3 \\ \text{QuotientSpace}[9, 1] \to 9 & \text{QuotientSpace}[13, 2] \to 12 \\ \text{QuotientSpace}[63, 1] \to 63 & \text{QuotientSpace}[82, 8] \to 75 \end{pmatrix}$$

Other runs

```
Analyze[{"R3b", "TC", "R1"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[4, 5] \to 1 & \text{QuotientSpace}[4, 5] \to 1 \\ \text{QuotientSpace}[26, 51] \to 1 & \text{QuotientSpace}[30, 60] \to 2 \\ \text{QuotientSpace}[267, 777] \to 2 & \text{QuotientSpace}[297, 844] \to 4 \end{pmatrix}$$

```
Analyze[{"R3b", "TC"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[9, 7] \to 4 & \text{QuotientSpace}[11, 7] \to 6 \\ \text{QuotientSpace}[64, 97] \to 7 & \text{QuotientSpace}[75, 103] \to 13 \\ \text{QuotientSpace}[625, 1501] \to 12 & \text{QuotientSpace}[700, 1603] \to 25 \end{pmatrix}$$

```
Analyze[{"R3b", "TC", "R1", "Round"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[1, 3] \to 0 & \text{QuotientSpace}[1, 3] \to 0 \\ \text{QuotientSpace}[8, 29] \to 0 & \text{QuotientSpace}[9, 34] \to 0 \\ \text{QuotientSpace}[54, 253] \to 0 & \text{QuotientSpace}[63, 298] \to 0 \end{pmatrix}$$

```
Analyze[{"R3b", "TC", "Round"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[3, 3] \to 1 & \text{QuotientSpace}[4, 3] \to 2 \\ \text{QuotientSpace}[11, 15] \to 1 & \text{QuotientSpace}[15, 17] \to 3 \\ \text{QuotientSpace}[70, 185] \to 1 & \text{QuotientSpace}[85, 205] \to 4 \end{pmatrix}$$

```
Analyze[{"R3b", "R1", "Round"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[1, 3] \to 0 & \text{QuotientSpace}[1, 3] \to 0 \\ \text{QuotientSpace}[12, 29] \to 1 & \text{QuotientSpace}[13, 33] \to 1 \\ \text{QuotientSpace}[107, 297] \to 4 & \text{QuotientSpace}[122, 345] \to 5 \end{pmatrix}$$

```
Analyze[{"R3b", "R1"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[4, 4] \to 2 & \text{QuotientSpace}[4, 4] \to 2 \\ \text{QuotientSpace}[32, 37] \to 7 & \text{QuotientSpace}[36, 43] \to 9 \\ \text{QuotientSpace}[460, 749] \to 42 & \text{QuotientSpace}[504, 827] \to 51 \end{pmatrix}$$

```
Analyze[{"R3b", "R1", "Descending"}] // MatrixForm
```

$$\begin{pmatrix} \text{QuotientSpace}[1, 1] \to 0 & \text{QuotientSpace}[1, 1] \to 0 \\ \text{QuotientSpace}[3, 3] \to 1 & \text{QuotientSpace}[4, 4] \to 1 \\ \text{QuotientSpace}[22, 23] \to 6 & \text{QuotientSpace}[28, 30] \to 7 \end{pmatrix}$$